

高等学校教材

高等学校计算机实验教学示范中心精品教材

Visual Basic.NET 程序设计基础 实验教程

薛梅 主编

巩艳华 刘莉 李洪国 副主编



高等教育出版社
HIGHER EDUCATION PRESS

高等学校教材
高等学校计算机实验教学示范中心精品教材

Visual Basic.NET 程序设计基础 实验教程

Visual Basic.NET Chengxu Sheji Jichu Shiyao Jiaocheng

薛 梅 主编
巩艳华 刘 莉 李洪国 副主编



高等教育出版社·北京
HIGHER EDUCATION PRESS BEIJING

内容提要

本书是《Visual Basic.NET 程序设计基础》的配套实验教材。

本书结合课程教学特点,分为三个部分。第一部分为实验指导,包括18个实验,实验内容与主教材内容配合,每个实验包括实验目的、基础训练和能力提高三个部分。第二部分为综合练习,共5套试题,内容涵盖主教材的知识点,以试题的形式对学习者的基础知识和综合能力进行考查。第三部分为习题参考答案,是对主教材每章习题的解答,便于学生在学习过程中参考。

本书紧密配合主教材,内容丰富,循序渐进,可以作为高等学校 Visual Basic.NET 程序设计课程的辅助教材,也可以作为广大应用程序开发者的入门参考书。

图书在版编目(CIP)数据

Visual Basic.NET程序设计基础实验教程 / 薛梅主编.
— 北京: 高等教育出版社, 2013.2
ISBN 978-7-04-036781-2

I. ①V… II. ①薛… III. ①
BASIC语言—程序设计—高等学校—教材 IV. ①TP312

中国版本图书馆CIP数据核字(2013)第009008号

策划编辑 时 阳 责任编辑 耿 芳 封面设计 赵 阳 版式设计 王艳红
责任校对 李大鹏 责任印制 朱学忠

出版发行	高等教育出版社	咨询电话	400-810-0598
社 址	北京市西城区德外大街4号	网 址	http://www.hep.edu.cn
邮政编码	100120		http://www.hep.com.cn
印 刷	保定市中国画美凯印刷有限公司	网上订购	http://www.landaco.com
开 本	787mm×1092mm 1/16		http://www.landaco.com.cn
印 张	9.25	版 次	2013年2月第1版
字 数	220千字	印 次	2013年2月第1次印刷
购书热线	010-58581118	定 价	13.40元

本书如有缺页、倒页、脱页等质量问题,请到所购图书销售部门联系调换
版权所有 侵权必究
物 料 号 36781-00

前 言

本书是《Visual Basic.NET 程序设计基础》一书的配套实验教材，全书分为“实验指导”、“综合练习”和“习题参考答案”三部分。内容与主教材相互补充，有效地扩展了主教材内容的深度和广度。

“实验指导”包括 18 个实验，实验内容与主教材内容配合，每个实验包括实验目的、基础训练和能力提高三个部分。其中基础训练分为程序验证、程序完善、程序改错三种类型，以不同的形式训练了学习者的编程能力。能力提高是在经过基础训练、掌握基础知识之后的知识扩展和补充。

“综合练习”共 5 套试题，题型包括选择题和程序填充两种，内容涵盖了主教材的知识点，以试题的形式对学习者的基础知识和综合能力进行考查，使得学习者能够综合掌握编程基础知识，加强分析问题和解决问题的能力。

“习题参考答案”是对主教材每章习题的解答，便于学生自检自测。

本书结构清晰，“实验指导”操作步骤详细，以多种形式加深学习者对知识的理解和掌握；“综合练习”知识点涵盖全面；“习题参考答案”解析详尽，可以作为高等学校计算机基础课教程“Visual Basic.NET 程序设计基础”的实验指导教材，也可以作为 Visual Basic.NET 系统开发的指导用书。

本书由薛梅担任主编，巩艳华、刘莉、李洪国担任副主编。高等教育出版社为本书的策划、出版做了大量工作，在此表示衷心的感谢。

由于时间仓促以及编者水平有限，书中难免存在疏漏之处，恳请读者批评指正。

作者 E-mail: xiaoxue619@163.com。

编 者

2012 年 9 月

目 录

第 1 部分 实验指导	1	1.8.2 基础训练	37
1.1 初识 Visual Basic.NET	1	1.8.3 能力提高	42
1.1.1 实验目的	1	1.9 二维数组程序设计	42
1.1.2 基础训练	1	1.9.1 实验目的	42
1.1.3 能力提高	4	1.9.2 基础训练	42
1.2 标准控件应用 (1)	4	1.9.3 能力提高	49
1.2.1 实验目的	4	1.10 动态数组及结构数组的应用	49
1.2.2 基础训练	4	1.10.1 实验目的	49
1.2.3 能力提高	9	1.10.2 基础训练	49
1.3 标准控件应用 (2)	10	1.10.3 能力提高	56
1.3.1 实验目的	10	1.11 Sub 与 Function 过程程序设计	57
1.3.2 基础训练	10	1.11.1 实验目的	57
1.3.3 能力提高	14	1.11.2 基础训练	57
1.4 常量、变量、运算符、函数及 表达式应用	15	1.11.3 能力提高	61
1.4.1 实验目的	15	1.12 参数传递及递归过程	62
1.4.2 基础训练	15	1.12.1 实验目的	62
1.4.3 能力提高	19	1.12.2 基础训练	62
1.5 分支结构程序设计	20	1.12.3 能力提高	67
1.5.1 实验目的	20	1.13 界面设计 (1)	68
1.5.2 基础训练	20	1.13.1 实验目的	68
1.5.3 能力提高	26	1.13.2 基础训练	68
1.6 循环结构	26	1.13.3 能力提高	76
1.6.1 实验目的	26	1.14 界面设计 (2)	76
1.6.2 基础训练	26	1.14.1 实验目的	76
1.6.3 能力提高	31	1.14.2 基础训练	77
1.7 控制结构综合应用	31	1.14.3 能力提高	82
1.7.1 实验目的	31	1.15 文件访问	82
1.7.2 基础训练	31	1.15.1 实验目的	82
1.7.3 能力提高	36	1.15.2 基础训练	83
1.8 一维数组程序设计	36	1.15.3 能力提高	86
1.8.1 实验目的	36	1.16 数据库应用	86
		1.16.1 实验目的	86

1.16.2 基础训练	86	2.3.2 参考答案	115
1.16.3 能力提高	89	2.4 综合练习 4	116
1.17 面向对象程序设计基础	89	2.4.1 试题	116
1.17.1 实验目的	89	2.4.2 参考答案	119
1.17.2 基础训练	90	2.5 综合练习 5	119
1.17.3 能力提高	96	2.5.1 试题	119
1.18 综合训练	96	2.5.2 参考答案	124
1.18.1 实验目的	96	第 3 部分 习题参考答案	125
1.18.2 基础训练	97	3.1 Visual Basic.NET 概述	125
第 2 部分 综合练习	102	3.2 面向对象的可视化编程基础	125
2.1 综合练习 1	102	3.3 语言基础	126
2.1.1 试题	102	3.4 基本控制结构	126
2.1.2 参考答案	107	3.5 数组	129
2.2 综合练习 2	107	3.6 过程	132
2.2.1 试题	107	3.7 用户界面设计	133
2.2.2 参考答案	111	3.8 文件访问	138
2.3 综合练习 3	111	3.9 ADO.NET 与数据库	139
2.3.1 试题	111	3.10 面向对象程序设计基础	139

第1部分 实验指导

1.1 初识 Visual Basic.NET

1.1.1 实验目的

1. 熟悉 Visual Basic.NET（以下简称 VB.NET）程序开发环境，了解 VB.NET 应用程序的基本框架结构。
2. 掌握 Windows Form 的事件处理机制。
3. 初步掌握 VB.NET 的基本编程方法和技巧。

1.1.2 基础训练

1. 程序验证

【实验 1】

【要求】编写一个程序，用户单击“中文”按钮后，标签显示“世界，你好！”。单击“英文”按钮时，标签显示“Hello World!”。窗体界面如图 1.1.1 所示。

【分析】通过在代码中使用赋值语句来改变标签内容。

【操作步骤】

(1) 新建项目

在 Visual Studio 2008 中新建一个名为“中英文显示”的项目，在名为 Form1.vb 的窗体上添加两个命令按钮（Button）和一个标签（Label）。窗体布局如图 1.1.1 所示。

(2) 控件属性

各控件属性设置如表 1.1.1 所示（其他属性为默认值）。



图 1.1.1 实验 1 窗体界面

表 1.1.1 实验 1 控件属性表

控 件 名 称	属 性 名	值
Form1	Text	中英文显示
Label1	AutoSize	True
	BackColor	Transparent（透明）
	Font	宋体，14.25 pt
	BorderStyle	FixedSingle
	ForeColor	HotTrack

续表

控 件 名 称	属 性 名	值
Button1	Text	中文
	Font	宋体, 14.25 pt
Button2	Text	英文
	Font	宋体, 14.25 pt

(3) 代码添加

Public Class Form1

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click
    Label1.Text = "世界，你好！"
```

```
End Sub
```

```
Private Sub Button2_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button2.Click
    Label1.Text = "Hello World!"
```

```
End Sub
```

End Class

(4) 运行程序

按 F5 键或者单击工具栏中的 ▶ 按钮，运行该项目。运行结果如图 1.1.2 所示。

【实验 2】

【要求】编写一个简单程序，要求单击命令按钮来显示或隐藏标签。窗体界面如图 1.1.3 所示。




图 1.1.2 实验 1 运行界面




图 1.1.3 实验 2 窗体界面

【分析】通过在代码中使用显示/隐藏语句来实现控件的显示/隐藏。

【操作步骤】

(1) 新建项目

在 Visual Studio 2008 中新建一个名为“显示隐藏”的项目，在名为 Form1.vb 的窗体上添加两个命令按钮（Button）和一个标签（Label）。窗体布局如图 1.1.3 所示。

(2) 控件属性

各控件属性设置如表 1.1.2 所示（其他属性为默认值）。

表 1.1.2 实验 2 控件属性表

控 件 名 称	属 性 名	值
Form1	Text	显示隐藏
Button1	Text	显示
Button2	Text	隐藏
Label1	Text	VB.NET 程序设计
	Visible	False

(3) 代码添加

Public Class Form1

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click  
    Label1.Show()  
End Sub
```

```
Private Sub Button2_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button2.Click  
    Label1.Hide()  
End Sub
```

End Class

End Class

(4) 运行程序

按 F5 键或者单击工具栏中的 ▶ 按钮，运行该项目。运行结果如图 1.1.4 所示。




图 1.1.4 实验 2 运行界面

2. 程序完善

【实验 3】

【要求】通过单击或双击窗体改变窗体的标题。单击窗体，标题显示“这是单击事件”，双击窗体，标题显示“这是双击事件”。程序运行结果如图 1.1.5 所示。




图 1.1.5 实验 3 运行界面

【分析】单击事件是 Click，双击事件是 DoubleClick，窗体标题是 Text 属性。

【程序代码】

Public Class Form1

```
Private Sub Form1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Me.Click  
    ' 单击效果
```

End Sub

```
Private Sub Form1_DoubleClick(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles _  
Me.DoubleClick
```

```
    ' 双击效果
```

End Sub

End Class

3. 程序改错

【实验 4】

【要求】编写程序，通过输入圆的半径，单击“圆的面积”按钮，求圆形的面积。程序运行

结果如图 1.1.6 所示。

注意：改错时不能删除语句，也不能增加语句，但可以移动语句的位置。

【分析】圆面积的计算公式：圆面积= $\pi * r * r$ 。

【有错误的程序代码】

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System._
EventArgs) Handles Button1.Click
    TextBox2.Text =  $\pi * \text{TextBox1.Text} * \text{TextBox1.Text}$ 
End Sub
```




图 1.1.6 实验 4 运行界面

1.1.3 能力提高

【练习题】

1. 单击“开始”按钮，在文本框中显示红色“Hello world!”；单击“清屏”按钮，文本框中不显示任何信息；单击“结束”按钮，退出运行。程序运行结果如图 1.1.7 所示。

2. 向两个文本框中分别输入两个数，单击“计算”按钮，在第三个文本框中显示计算结果，单击“退出”按钮，退出应用程序。程序运行结果如图 1.1.8 所示。



图 1.1.7 练习题 1 运行界面




图 1.1.8 练习题 2 运行界面

提示：Val() 函数表示把输入文本框的内容转换成数值；Str() 函数表示将数值转换成字符串。

1.2 标准控件应用 (1)

1.2.1 实验目的

1. 掌握 VB.NET 中标准控件的使用。
2. 掌握标准控件的事件处理机制。
3. 掌握 VB.NET 标准控件的基本编程方法。

1.2.2 基础训练

1. 程序验证

【实验 1】

【要求】编写一个简单程序，要求单击命令按钮来显示文字、改变文字和改变颜色。窗体界面如图 1.2.1 所示。

【分析】在代码中改变 Label1 的值和颜色。




图 1.2.1 实验 1 窗体界面

【操作步骤】

(1) 新建项目

在 Visual Studio 2008 中新建一个名为“文字变化”的项目，在名为 Form1.vb 的窗体上添加 3 个命令按钮 (Button) 和一个标签 (Label)。窗体布局如图 1.2.1 所示。

(2) 控件属性

各控件属性设置如表 1.2.1 所示 (其他属性为默认值)。

表 1.2.1 实验 1 控件属性表

控 件 名 称	属 性 名	值
Form1	Text	文字变化
Button1	Text	显示文字
Button2	Text	改变文字
Button3	Text	改变颜色

(3) 代码添加

Public Class Form1

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click  
    Label1.Text = "这是我的第一个 VB.NET 程序！"
```

```
End Sub
```

```
Private Sub Button2_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button2.Click  
    Label1.Text = "怎么样？还不错吧！"
```

```
End Sub
```


```
Private Sub Button3_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button3.Click  
    Label1.ForeColor = Color.Blue
```

```
End Sub
```


End Class

(4) 运行程序


按 F5 键或者单击工具栏中的 ▶ 按钮，运行该项目。运行结果如图 1.2.2 所示。



(a) 显示文字效果



(b) 改变文字效果



(c) 改变颜色效果

图 1.2.2 实验 1 运行界面

【实验 2】

【要求】编写一个简单程序，要求从文本框中输入两个数，单击命令按钮来计算这两个数相加之和与相乘之积。窗体界面如图 1.2.3 所示。

【分析】通过 Val() 函数将 TextBox1 和 TextBox2 的内容转换成数值，实现相加和相乘来改变标签内容。

【操作步骤】

(1) 新建项目

在 Visual Studio 2008 中新建一个名为“相加相乘”的项目，在名为 Form1.vb 的窗体上添加一个命令按钮 (Button)、两个文本框 (TextBox) 和两个标签 (Label)。窗体布局如图 1.2.3 所示。



图 1.2.3 实验2 窗体界面

(2) 控件属性

各控件属性设置如表 1.2.2 所示 (其他属性为默认值)。

表 1.2.2 实验2 控件属性表

控 件 名 称	属 性 名	值
Form1	Text	相加相乘
Button1	Text	结果

(3) 代码添加

```
Public Class Form1
```

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click
```

```
Label1.Text = Val(TextBox1.Text) + Val(TextBox2.Text)
```

```
Label2.Text = Val(TextBox1.Text) * Val(TextBox2.Text)
```

```
End Sub
```

```
End Class
```

(4) 运行程序

按 F5 键或者单击工具栏中的 按钮，运行该项目。在文本框中分别输入“12”和“30”，单击“结果”按钮，程序运行结果如图 1.2.4 所示。

【实验 3】

【要求】编写一个人民币与美元兑换的程序。在“人民币”或“美元”文本框中输入要兑换的钱款，在“兑换比率”框输入人民币兑换成美元的兑换比，单击所需兑换的按钮，进行相应的兑换，并在文本框中显示结果；单击“清屏”按钮，清除文本框中的内容；单击“结束”按钮，停止程序运行。窗体界面如图 1.2.5 所示。



图 1.2.4 实验2 运行界面




图 1.2.5 实验3 窗体界面

【分析】通过输入“兑换比率”来修改数值。

【操作步骤】

(1) 新建项目

在 Visual Studio 2008 中新建一个名为“兑换”的项目，在名为 Form1.vb 的窗体上添加 4 个命令按钮 (Button)、3 个文本框 (TextBox) 和 3 个标签 (Label)。窗体布局如图 1.2.5 所示。

(2) 控件属性

各控件属性设置如表 1.2.3 所示（其他属性为默认值）。

表 1.2.3 实验 3 控件属性表

控 件 名 称	属 性 名	值
Form1	Text	兑换
Label1	Text	人民币
Label2	Text	兑换比率
Label3	Text	美元
Button1	Text	清屏
Button2	Text	¥ → \$
Button3	Text	¥ ← \$
Button4	Text	结束

(3) 代码添加

```
Public Class Form1
    Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click
        TextBox1.Clear()
        TextBox2.Clear()
        TextBox3.Clear()
    End Sub
    Private Sub Button2_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button2.Click
        TextBox3.Text = Str(Val(TextBox1.Text) / Val(TextBox2.Text))
        ' Val()函数将输入文本框的内容转换成数值, Str()函数将数值转换成字符串
    End Sub
    Private Sub Button3_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button3.Click
        TextBox1.Text = Str(Val(TextBox3.Text) * Val(TextBox2.Text))
    End Sub
    Private Sub Button4_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button4.Click
        End
    End Sub
End Class
```

(4) 运行程序

按 F5 键或者单击工具栏中的 按钮，运行该项目。在“人民币”文本框中输入“100”，在“兑换比率”文本框中输入“8.29”，单击“¥ → \$”按钮，程序运行结果如图 1.2.6 所示。

2. 程序完善

【实验 4】

【要求】编写 3 个事件过程，通过装入、单击或双击窗体改变显示效果，使用 Load 事件、Click 事件和 DoubleClick 事件。程序运行界面如图 1.2.7 所示。




图 1.2.6 实验 3 运行界面




图 1.2.7 实验 4 运行界面

【分析】

(1) 在窗体装入时，窗体的标题栏显示“装入窗体”，并以图片作为背景平铺窗体，窗体边框为“Fixed3D”，如图 1.2.7 (a) 所示。

(2) 单击窗体，标题栏显示对应的文字，图片大小正好充满窗体，如图 1.2.7 (b) 所示。

(3) 双击窗体，窗体无最大化按钮和最小化按钮，标题栏显示对应的文字，无图片，如图 1.2.7 (c) 所示。

【程序代码】

Public Class Form1

```
Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load
    Me.FormBorderStyle = Windows.Forms.FormBorderStyle.Fixed3D
```

' Load 事件窗体名称

```
Me.BackgroundImage = Image.FromFile("qq.jpg") ' 图片与应用程序在同一文件夹
```

```
End Sub
```

```
Private Sub Form1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Me.Click
```

```
Me.Text = "单击窗体"
```

```
Me.Size = New Size(200 + 10, 200 + 40)
```

' 窗体与图片大小相同，10、40 为边框线和标题栏高

' 事件单击时装载的图片

```
End Sub
```

```
Private Sub Form1_DoubleClick(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles _
```

```
Me.DoubleClick
```

```
Me.Text = "双击窗体"
```

' 窗体双击时的效果

```
Me.BackgroundImage = Nothing
```

```
End Sub
```

```
End Class
```

3. 程序改错

【实验 5】

【要求】编写程序实现两个数据的交换。向“交换前”的两个文本框输入数据，单击“交换

数据”按钮，实现数据交换。程序运行结果如图 1.2.8 所示。




图 1.2.8 实验 5 运行界面

注意：改错时不能删除语句，也不能增加语句，但可以移动语句的位置。

【分析】两个数据交换，需要借助第三个变量。

【有错误的程序代码】

```
Public Class Form1
```

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click
```

```
Dim x As Double, y As Double
```

```
x = Val(TextBox1.Text)
```

```
y = Val(TextBox2.Text)
```

```
' 交换数据
```

```
x = y
```

```
y = x
```

```
TextBox3.Text = Str(x)
```

```
TextBox4.Text = Str(y)
```

```
End Sub
```

```
End Class
```

1.2.3 能力提高

【练习题】

1. 编写一个显示输入字符串的程序。运行结果如图 1.2.9 所示。

2. 编写一个实现标签移动的程序。要求：单击一次“手动”按钮，标签向下移动 50 像素。

运行结果如图 1.2.10 所示。




图 1.2.9 练习题 1 运行界面




图 1.2.10 练习题 2 运行界面

提示：标签向下移动即标签的 Top 属性增加 50 像素，若移动位置超过窗体高度，则再从 0 开始移动。

1.3 标准控件应用 (2)

1.3.1 实验目的

1. 掌握 VB.NET 中标准控件的使用。
2. 掌握标准控件的事件处理机制。
3. 掌握 VB.NET 标准控件的基本编程方法。

1.3.2 基础训练

1. 程序验证

【实验 1】

【要求】向窗体添加 3 个对象：一个标签、一个文本框和一个按钮。单击按钮，在文本框中显示“欢迎光临”文字。窗体界面如图 1.3.1 所示。

【分析】通过在代码中使用赋值语句来改变文本框的内容。

【操作步骤】

(1) 新建项目

在 Visual Studio 2008 中新建一个名为“文字显示”的项目，在名为 Form1.vb 的窗体上添加一个命令按钮 (Button)、一个文本框 (TextBox) 和一个标签 (Label)。窗体布局如图 1.3.1 所示。

(2) 控件属性

各控件属性设置如表 1.3.1 所示 (其他属性为默认值)。

表 1.3.1 实验 1 控件属性表

控 件 名 称	属 性 名	值
Button1	Text	按钮
Label1	Text	标签
Text1	Text	文本框

(3) 代码添加

```
Public Class Form1
```

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click  
    TextBox1.Text = "欢迎光临"
```

```
End Sub
```

```
End Class
```

(4) 运行程序

按 F5 键或者单击工具栏中的 按钮，运行该项目。运行结果如图 1.3.2 所示。

【实验 2】

【要求】输入矩形的宽和高，实现求矩形的面积。窗体界面如




图 1.3.1 实验 1 窗体界面




图 1.3.2 实验 1 运行界面

图 1.3.3 所示。

【分析】矩形的面积计算公式：矩形的宽×矩形的高，通过在代码中使用赋值语句来改变标签内容。

【操作步骤】

(1) 新建项目

在 Visual Studio 2008 中新建一个名为“矩形面积”的项目，在名为 Form1.vb 的窗体上添加两个命令按钮 (Button)、3 个文本框 (TextBox) 和 3 个标签 (Label)。窗体布局如图 1.3.3 所示。

(2) 控件属性

各控件属性设置如表 1.3.2 所示 (其他属性为默认值)。




图 1.3.3 实验 2 窗体界面

表 1.3.2 实验 2 控件属性表

控 件 名 称	属 性 名	值
Form1	Text	矩形面积
Label1	Text	矩形的宽
Label2	Text	矩形的高
Label3	Text	矩形的面积
Button1	Text	计算
Button2	Text	退出

(3) 代码添加

```
Public Class Form1
```

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click  
    TextBox3.Text = Str(Val(TextBox1.Text) * Val(TextBox2.Text))
```

```
End Sub
```

```
Private Sub Button2_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button2.Click  
    End
```

```
End Sub
```

```
End Class
```

(4) 运行程序

按 F5 键或者单击工具栏中的 按钮，运行该项目。在“矩形的宽”文本框中输入“10”，在“矩形的高”文本框中输入“20”，单击“计算”按钮，程序运行结果如图 1.3.4 所示。



图 1.3.4 实验 2 运行界面